

WCS in PICARD-SOL images

Thierry Corbard
Laboratoire Lagrange/ Observatoire de la Côte d'Azur

Created July 10, 2014 / Last updated Dec 9 2015

1 Introduction

A basic and standard way of expressing coordinate systems in FITS files were first described by Wells et al. (1981) together with the definition of the FITS format. The World Coordinate System (WCS) (Greisen & Calabretta, 2002) is an extension which allows to describe the system of projection used for spherical coordinates. This extension has then been adapted for solar coordinates by Thompson (2006) who has also developed a set of WCS tools for the SolarSoft IDL library (Thompson, 2010). This short note shows how this has been implemented for PICARD-SOL images and gives few examples illustrating the use of the WCS SolarSoft library¹ to analyze PICARD-SOL images.

2 WCS keywords in PICARD-SOL image header

The helioprojective-cartesian coordinates are given in the headers of level 1 PICARD-SOL images using the standard WCS FITS keywords.

Figure 1 gives an example of header with non-standard keywords commonly used to describe solar images. OBS_L0 and OBS_B0 give the Carrington heliographic position of the observer, DSUN his distance from the center of the sun in astronomical unit. QXCENTER, QYCENTER and SOLAR_P give respectively the coordinates of the solar disk center in fractional pixels² and the solar position angle.

```
OBS_L0 = 338.23 / [deg] Heliographic Longitude of the observer
OBS_B0 = 4.87 / [deg] Heliographic Latitude of the observer
DSUN = 1.01610908369640 / [au] Distance between Calern and Sun Center
SOLAR_P = 6.1663 / [deg] Position angle of the solar north pole
QXCENTER= 1050.47 / [1-index px] Quick look x-center
QYCENTER= 1029.60 / [1-index px] Quick look y-center
```

Figure 1: Non standard keywords used in PICARD-SOL level 0 (N0) FITS files

Figure 2 show how this is translated into standard WCS keywords. Because PICARD-SOL images are obtained from ground, the solar north pole is not on the CCD y-axis. The P-angle gives the angular distance between the directions of the solar north and the celestial north which, for an equatorial mount, is given by the CCD y-axis. The WCS rotation matrix $PC_{i,j}$ is therefore built from the knowledge of this angle³:

```
PC1_1 = cos(SOLAR_P)  PC1_2 = sin(SOLAR_P)
PC2_1 = - sin(SOLAR_P) PC2_2 = cos(SOLAR_P)
```

¹ <http://sohowww.nascom.nasa.gov/solarsoft/gen/idl/wcs/>

² FITS convention for fractional pixels is that the center of the lower left pixel is at (1,1), the lower left corner of the lower left pixel is at (0.5,0.5). [1-index pix] refers to this convention.

³ Note that this is valid assuming that the plate scale is the same in both direction (CDELTA1=CDELTA2)

```

WCSNAME = 'Helioprojective-Cartesian' / Heliocentric with Gnomonic projection
CTYPE1 = 'HPLN-TAN' / theta_x from apparent disk center
CTYPE2 = 'HPLT-TAN' / theta_y from apparent disk center
CRPIX1 = 1024.50 / [1-index px] x-coordinate of reference point
CRPIX2 = 1024.50 / [1-index px] y-coordinate of reference point
CRVAL1 = -27.9759 / [arcsec] theta_x at reference point
CRVAL2 = -2.4201 / [arcsec] theta_y at reference point
CUNIT1 = 'arcsec' / CELT1 and CVAL1 angles in arcsec
CUNIT2 = 'arcsec' / CELT2 and CVAL2 angles in arcsec
CDELTA1 = 1.0610 / [arcsec] x-platescale
CDELTA2 = 1.0610 / [arcsec] y-platescale
PC1_1 = 0.994214 / WCS transformation matrix
PC1_2 = 0.107415 / WCS transformation matrix
PC2_1 = -0.107415 / WCS transformation matrix
PC2_2 = 0.994214 / WCS transformation matrix
LONPOLE = 180.00 / [deg]
DSUN_OBS = 1.52007755320E+11 / [m] Observer s distance from solar center
HGLN_OBS = 0.00 / [deg] Stonyhurst heliographic long. of observer
HGLT_OBS = 4.87 / [deg] heliographic latitude of observer
CRLN_OBS = 338.23 / [deg] Carrington heliographic long. of observer

```

Figure 2: WCS standard keywords used in PICARD-SOL level 1 (N1) FITS files

The projection on the CCD plane is equivalent to a gnomonic projection (Calabretta & Greisen, 2002). This projection has its native north pole at a reference pixel, and the distance away from the pole increases as the tangent of the colatitude. Its WCS name is 'TAN'. The reference pixel represents the optical axis of the instrument which is not precisely determined. The best proxy for the optical axis, and therefore the reference point, is in our case the center of the CCD (rather than the center of the solar disk). This leads to the values of $CRPIX_i = (NAXIS_i + 1) / 2$. $CRVAL_i$ keywords then give the coordinates of the solar disk center relative to the reference point. $CUNIT_i$ keywords give the unit of these coordinates (arcsec) and $CDELTA_i$ keywords give the instrument plate-scale (Meftah et al., 2014). From the knowledge of $XCENTER$ and $YCENTER$, the [1-index px] pixel position of the Sun center, we get:

$$\begin{aligned}
CRVAL1 &= CDELTA1 * (PC1_1 * (CRPIX1 - QXCENTER) + PC1_2 * (CRPIX2 - QYCENTER)) \\
CRVAL2 &= CDELTA2 * (PC2_1 * (CRPIX1 - QXCENTER) + PC2_2 * (CRPIX2 - QYCENTER))
\end{aligned}$$

The keyword $LONPOLE$ represents the longitude in the native coordinate system of the celestial system's north pole (Calabretta & Greisen, 2002). It is always 180° in our case.

The WCS standard for the Carrington heliographic coordinates of the observer are $HGLT_OBS$ and $CRLN_OBS$. $HGLN_OBS$ gives the Stonyhurst heliographic longitude which is by definition 0 for an observer on earth. Finally, the distance between the observer and the center of the Sun is given by $DSUN_OBS$ in meters.

3 Examples using the SolarSoft WCS routines in IDL

3.1 Pixel coordinates of the solar disk center using standard WCS keywords

```
file='PIC_SOD2_N1_RS_WL607DL_20120720_1103_v01.fits.gz'  
im=readfits(file,header) ;Read file and header  
  
wcs=fitshead2wcs(header) ;Extract WCS structure from header  
  
center=wcs_get_pixel(wcs,[0,0]) ;Position of Sun center [0-index pix]4  
  
print, center ;Gives back the coordinates of the solar disk center  
;(with IDL convention for fractional pixel)4  
1049.47 1028.60 ;=QXCENTER-1 QYCENTER-1 (cf. Figure 1)
```

3.2 Limb Darkening Function at the solar north using standard WCS keywords.

```
coord=wcs_get_coord(wcs) ;Get helioprojective cartesian coordinates of all image pixels  
  
wcs_convert_from_coord,wcs,coord,'hpr',phi,rho,/zero_center ;Convert to helioprojective  
;radial ('hpr') coordinates  
  
polar=where(phi gt -0.5 and phi lt 0.5) ;Select position angle close to solar north (phi=0)  
  
plot,rho[polar]*3600.D0,im[polar],ps=3,xtitle='arcsec',title=file ;Plot the radial profile
```




Figure 3: radial profile close to the solar north ($\pm 0.5^\circ$) for a PICARD-SOL image at 607 nm obtained by converting the helioprojective cartesian coordinates of the image pixels into helioprojective radial coordinates using WCS standard keywords.

⁴ IDL convention for fractional pixels is that the center of the lower left pixel is at (0,0), the lower left corner of the lower left pixel is at (-0.5,-0.5). [0-index pix] refers to this convention.

3.3 Overlay of Carrington heliographic longitude and latitude lines

Figure 4 illustrates that standard WCS tools can be used to overplot the heliographic grid on a PICARD-SOL image (CRLN_OBS= 208.40).

```
;read image and header  
file='PIC_SOD2_N1_RS_WL393DL_20120730_0636_v01.fits.gz'  
im=readfits(file,header)  
  
;Get helioprojective cartesian coordinates  
wcs=fitshead2wcs(header)  
coord=wcs_get_coord(wcs)  
  
;Convert to Carrington heliographic coordinates ('hg' /carrington)  
wcs_convert_from_coord,wcs,coord,'hg',lon,lat,/CARRINGTON,/pos_long  
  
;display the result (using The "IDL Coyote Library" (D. Fanning, 2000))  
cgimage,im,/keep_aspect,ct=1,/save  
cgcontour,lat,/overplot,levels=[-60,-30,0,30,60],color='green'  
cgcontour,lon,/overplot,levels=wcs.position.crln_obs+[-60,-30,0,30,60],color='green'
```




Figure 4: PICARD-SOL 393 nm image with Carrington heliographic longitude and latitude overlaid using only WCS standard keywords and SolarSoft tools.

3.4 Rotation of the image and overlay of Carrington longitude and latitude lines

Figure 5 illustrates that standard WCS tools (here **hrot** from the IDL Astronomy Library (Landsman, 1993)) can be used to rotate the image from the P-angle and update the WCS keywords. The WCS keywords being updated, the same procedure as in previous section can then be used to overlay heliographic coordinates on the rotated image.

```
;read image and header
file='PIC_SOD2_N1_RS_WL393DL_20120730_0636_v01.fits.gz'
im=readfits(file,header)

;Get helioprojective cartesian coordinates
wcs=fitshead2wcs(header)

; Get center of rotation (Solar disk center)
center=wcs_get_pixel(wcs,[0,0])

; Get P-angle
;solar_P=sxpar(header,'SOLAR_P') ; Would gives 10.26° but this is not standard WCS
P=wcs_get_pixel(wcs,[0,400]) ;Pixel coordinates of a point P located 400 arcsec up on the
central meridian
d=P-center
solar_P=-atan(d[0],d[1])*180.D0/!DPI ;Get the angle between solar center and P

; Rotate / recenter image and update WCS keywords in header using hrot
hrot,im,header,-1,-1,solar_P,center[0],center[1],2

;Get new helioprojective cartesian coordinates
wcs=fitshead2wcs(header)
coord=wcs_get_coord(wcs)

;Convert to Carrington heliographic coordinates ('hg' /carrington)
wcs_convert_from_coord,wcs,coord,'hg',lon,lat,/CARRINGTON,/pos_long

;display the result
cgimage,im,/keep_aspect,ct=1,/save
cgcontour,lat,/overplot,levels=[-60,-30,0,30,60],color='green'
cgcontour,lon,/overplot,levels=wcs.position.crln_obs+[-60,-30,0,30,60],color='green'

;Check the the new image is centered
center=wcs_get_pixel(wcs,[0,0])
print, center ; Gives 1023.5 1023.5 [0-index] i.e. 1024.5 1024.5 [1-index]

;Check that Solar North is up
P=wcs_get_pixel(wcs,[0,400])
d=P-center
print,-atan(d[0],d[1])*180.D0/!DPI ; Gives 1.7d-14 ~ 0
```




Figure 5: PICARD-SOL 393 nm image rotated from P-angle with Carrington heliographic longitude and latitude overlaid using only WCS standard keywords and SolarSoft tools.

References

- Calabretta, M. R. and Greisen, E. W. (2002). Representations of celestial coordinates in FITS. *A&A.*, 395, 1077–1122.
- Fanning, D., *IDL Programming Techniques*, 2nd Ed (2000). The Coyote Library is available at: http://www.idlcoyote.com/programs/zip_files/coyoteprograms.zip
- Greisen, E. W. and Calabretta, M. R. (2002). Representations of world coordinates in FITS. *A&A.*, 395, 1061–1075.
- Landsman, W. B 1993 *The IDL Astronomy Library in Astronomical Data Analysis Software and Systems II*, A.S.P. Conference Series, Vol. 52, ed. R. J. Hanisch, R. J. V. Brissenden, and Jeannette Barnes, p. 246., <http://idlastro.gsfc.nasa.gov/>
- Meftah M., Corbard T., Irbah A., et al. (2014). Ground-based measurements of the solar diameter during the rising phase of solar cycle 24. *A&A* 569, A60
- Thompson, W. T. (2006). Coordinate systems for solar image data. *Astron. Astrophys.*, 449, 791–803.
- Thompson, W. T. (2010). *The SolarSoft WCS Routines: A Tutorial*. http://hesperia.gsfc.nasa.gov/ssw/gen/idl/wcs/wcs_tutorial.pdf, <http://sohowww.nascom.nasa.gov/solarsoft/gen/idl/wcs/>
- Wells, D. C., Greisen, E. W., and Harten, R. H. (1981). FITS - a Flexible Image Transport System. *Astron. Astrophys. Supp.*, 44, 363.